

430eForth43n6vis – VOCs and ITEMS HOWTO – Creating a DS1621 Library

Manfred Mahlow

Zusammenfassung

Some decades ago, when I first discovered Forth and became familiar with the vocabulary concepts of Fig-Forth and F83, I was missing a way to bind a vocabulary (a namespace) to a word representing data, to create data types or simple objects.

Over the years I found a way to do it without using state-smart words, first for PC Forth Systems and during the last years also for Flash based MCU Forth Systems. VOCs and ITEMS [2] are the core elements of the final concept and 430eForth43n6vis [3] is at present the smallest Forth System fully supporting both. This HOWTO should give an idea, how to use VOCs and ITEMS (on a MSP 430 LaunchPad) to create a library.

A Library for the DS1621 Digital Thermometer and Thermostat

The DS1621 [1] provides 9-bit readings, which indicate the temperature of the device. The device has three temperature registers, T for the last measured temperature, TL and TH for thermostat settings. Data is read/written via a 2-wire serial interface (open drain I/O lines). Three address inputs allow to address 8 devices per bus. The base address is 48h. The device operation is controlled by an 8-bit configuration register. Two conversion modes are supported, a 1SHOT mode and a continous mode and the thermostat status is available as a digital output.

1 I2C (2-WIRE) Library required

The DS1621 is an I2C Bus device. So some words from an I2C Library are required for read and write access.

```
1 \ DS1621 Digital Thermometer and Thermostat with I2C Bus Interface      MM-180804
2 \
3 \ Up to 8 DS1621 devices can be connected to a single I2C bus.
4 \
5
6 #require I2C      \ loading an I2C library  ( I2C Bus Master )
7
8 { The following words are used from the I2C library :
9
10   I2C ?ADDR ( u a1 — a2 )
11   Calculate the I2C slave address a2 from a given base address (a1) and a
12   modifier (u). Throw an error if u > 7.
13
14   I2C ?BUSY ( — )
15   Wait while the I2C Bus Master is busy (required when using I2C hardware).
16
17   I2C INIT ( — )
18   Initialise the I2C Bus interface.
19
20   I2C READ ( a +n sid — c1 .. cn )
21   Read +n bytes from address a of the I2C slave sid, sid is the slaves 7 bit
```

```

22     identifier or address. An ambiguous condition exists for n = 0.
23
24     I2C WRITE ( cn .. c1 +n sid — )
25     Write +n bytes to the I2C slave sid. sid is the slaves 7 bit id or addr. An
26     ambiguous condition exists for n = 0.
27 }

```

2 A Namespace for the DS1621 Library

A separate namespace is used for the library words. It's created with a vocabulary prefix (VOC), so that all library words can be accessed with this prefix.

```

29
30     ONLY FORTH DEFINITIONS    DECIMAL
31
32     VOC DS1621    \ the library namespace

```

3 DS1621 read/write access

Words to write to and read from a DS1621 device:

```

34     DS1621 DEFINITIONS    HEX
35
36     \ Return the I2C slave identifier of DS1621 device u ( u = [0-7] ).
37     : SID ( u — sid ) $48 I2C ?ADDR I2C ?BUSY I2C INIT ;
38
39     \ Read n bytes from address a of DS1621 device u ( u = [0-7] ).
40     : READ ( u a +n — c1 .. cn )
41       ROT DS1621 SID ( a +n sid ) I2C READ ;
42
43     \ Write n bytes to address a of DS1621 device u ( u = [0-7] ).
44     : WRITE ( cn — c1 u a n+1 — )
45       ROT DS1621 SID ( cn .. c1 n+1 sid — ) I2C WRITE ;

```

4 DS1621 CONFIG/STATUS Register

The DS1621 configuration and status is hold in an 8-bit register. It's the only 8-bit register in that device. So creating a data type for 8-bit DS1621 registers makes no sence here but using a vocabulary prefix makes sence.

```

47
48     DS1621 DEFINITIONS
49
50     VOC CONFIG    \ CONFIG/STATUS Register access
51
52     DS1621 CONFIG DEFINITIONS
53
54     \ Read the CONFIG/STATUS register of the DS1621 device u (u = [0-7]).
55     : READ ( u — c ) $AC 1 DS1621 READ ;
56
57     \ Write the byte c to the CONFIG register of the DS1621 device u (u = [0-7]).
58     : WRITE ( c u — ) $AC 2 DS1621 WRITE ;

```

5 DS1621 Temperature Registers

The DS1621 has three 16-bit temperature registers using a special format. So it makes sence, to use a VOC to create a data type for this registers.

5.1 Defining a data type for the temperature registers

```

61   DS1621 DEFINITIONS
62
63   VOC TR16 \ Temperature Register access
64
65   DS1621 TR16 DEFINITIONS
66
67   \ Read from the temperature register a of the DS1621 device u ( u = [0-7] ).
68   : READ ( u a — msb lsb ) 2 DS1621 READ ;
69
70   \ Write to the temperature register a of the DS1621 device u ( u = [0-7] ).
71   : WRITE ( msb lsb u a — ) >R >R SWAP R> R> 3 DS1621 WRITE ;
72
73   \ Read and print the temperature from register a of the DS1621 device u.
74   : ? ( u a — )
75     BASE @ >R DECIMAL DS1621 TR16 READ ( msb lsb )
76     IF [ CHAR 5 ] LITERAL ELSE [ CHAR 0 ] LITERAL THEN
77     <# HOLD [ CHAR . ] LITERAL HOLD ( S>D ) #S #> SPACE TYPE R> BASE ! ;
78
79   \ Note: # and #S expect single not double precision numbers in 430eForth.
80
81   \ Create an identifier for a DS1621 temperature register with address a.
82   : ITEM: ( "name" a — ) \ name ( — a , NS: DS1621 TR16 )
83     ITEM CONSTANT ;
84
85   DS1621 DEFINITIONS

```

5.2 Using the data type for the temperature registers

```

85   DS1621 DEFINITIONS
86
87   \ DS1621 Temperature Registers :
88   $AA DS1621 TR16 ITEM: T ( — a , NS: DS1621 TR16 )
89   $A1 DS1621 TR16 ITEM: TH ( — a , NS: DS1621 TR16 )
90   $A2 DS1621 TR16 ITEM: TL ( — a , NS: DS1621 TR16 )

```

6 Control and Monitoring

```

92   \ Start the temperature conversion at DS1621 device u ( u = [0-7] ).
93   \ Default on power up is the continous conversion mode.
94   : START ( u — ) $EE 1 DS1621 WRITE ;
95
96   \ Stop the tempeature conversion at DS1621 device u ( u = [0-7] ).
97   : STOP ( u — ) $22 1 DS1621 WRITE ;
98
99   \ Read the conversion status of DS1621 device u ( u = [0-7] ). Conversion
100  \ complete if f <> 0, in progress otherwise.
101  : DONE ( u — f ) DS1621 CONFIG READ 080 AND ;
102
103  FORTH DEFINITIONS DECIMAL
104
105  \ Last Revision: MM-191212 minor changes
106  \ MM-181120 converted to upper case
107  \ MM-180906 changed to use ITEMS
108  \ MM-180804 changed to used 430eForth43n1vi VOCs

```

7 Examples how to access a DS1621 Device

```

111
112  \ Some examples how to access a DS1621 sensor:
113
114  1. Direct access with the sensor id:
115
116  ( u — ) DS1621 <method>

```

```

117
118
119 2. Access with an context switching constant:
120   ( u — ) DS1621 ITEM CONSTANT TS
121
122   TS ( u — , NS: DS1621 ) <method>
123
124
125
126 3. Access with a context switching word defined with a colon definition:
127   DS1621 ITEM : TS ( — u ) ..... ;
128
129   TS ( — u , NS: DS1621 ) <method>
130
131
132
133 4. Using a sensor defining word:
134   DS1621 DEFINITIONS
135
136   : SENSOR ( "name" u — ) ITEM CONSTANT ;
137
138   FORTH DEFINITIONS
139
140   ( u — ) DS1621 SENSOR TS
141

```

8 The DS1621 Library in a Forth Terminal

8.1 ??

is the dictionary browser that shows the current search order, the compiler context, the data stack and the words of the top wordlist in the search order (not all words are shown here).

```

Terminal
Datei Bearbeiten Ansicht Suchen Terminal Hilfe
??
CONTEXT: FORTH  ROOT
CURRENT: FORTH
<sp
DS1621 ms us I2C END-CODE CODE RESET MARKER \ef \ipa SAVE ITEMS VOCS .VID COLD
STICKY VOC FORTH ROOT APP! HI VARIABLE CONSTANT CREATE :NN : ] ; OVERT $,n ." $"
ABORT" WHILE ELSE AFT THEN REPEAT IF AGAIN UNTIL NEXT BEGIN FOR LITERAL , IWRIT

```

DS1621 and I2C are the libraries that were loaded into the FORTH namespace.

8.2 VOCS

lists all vocabulary prefixes (all namespaces) that are defined in the dictionary.

```

VOCS
DS1621 TR16 DS1621 CONFIG DS1621 I2C FORTH ROOT ok

```

8.3 ITEMS

lists all context switching items that are defined in the dictionary.

```
ITEMS
DS1621 TR16 : DS1621 TL
DS1621 TR16 : DS1621 TH
DS1621 TR16 : DS1621 T
STICKY : ROOT ?? ok
```

?? is the dictionary browser defined in the ROOT namespace. It's a STICKY word, i.e. it does not change the actual search order.

8.4 DS1621 WORDS

```
DS1621 WORDS
DONE STOP START TL TH T TR16 CONFIG WRITE READ SID ok
```

8.5 DS1621 connected as device 0

```
0 DS1621 START ok
0 DS1621 T ? 20.5 ok
0 DS1621 ITEM CONSTANT TS1 ok
TS1 T ? 20.5 ok
19 0 TS1 TL WRITE    TS1 TL ? 19.0 ok
                     TS1 TL READ  SWAP . . 19 0 ok
21 0 TS1 TH WRITE    TS1 TH ? 21.0 ok
TS1 STOP ok
```

Literatur

- [1] DS1621 Data Sheet
<https://pdfserv.maximintegrated.com/en/ds/DS1621.pdf>
- [2] MANFRED MAHLOW, *VOCS, ITEMS und STICKY Words*, Forth Tagung 2019
https://wiki.forth-ev.de/lib/exe/fetch.php/events:ft2019:vocs_items_und_sticky_words.pdf
- [3] MANFRED MAHLOW, *Namespaces and Context Switching for a Tiny Forth*, VD 4/2019